<commit_before>{ stdenv, fetchgit, php }:

let
  libphutil = fetchgit {
    url    = "git://github.com/facebook/libphutil.git";
    rev    = "c4cb6d99c4a5903079350f85fcc71895c0a0ea14";
    sha256 = "a7293aac4fdcfbaead09ee3e6ffb54c5d100b07905b4006194067411061ff994";
  };
  arcanist = fetchgit {
    url    = "git://github.com/facebook/arcanist.git";
    rev    = "50caec620a8ed45c54323cb71fee72fd0d935115";
    sha256 = "dd18ed22375ad1ba058703952be0d339d9c93704e9d75dd7e4e6625236dfe9b0";
  };
in
stdenv.mkDerivation rec {
  name    = "arcanist-${version}";
  version = "20140530";

  src = [ arcanist libphutil ];
  buildInputs = [ php ];

  unpackPhase = "true";
  buildPhase = "true";
  installPhase = ''
    mkdir -p $out/bin $out/libexec
    cp -R ${libphutil} $out/libexec/libphutil
    cp -R ${arcanist}  $out/libexec/arcanist

    ln -s $out/libexec/arcanist/bin/arc $out/bin
  '';

  meta = {
    description = "Command line interface to Phabricator";
    homepage    = "http://phabricator.org";
    license     = stdenv.lib.licenses.asl20;
    platforms   = stdenv.lib.platforms.unix;
    maintainers = [ stdenv.lib.maintainers.thoughtpolice ];
  };
}
<commit_msg>arcanist: Fix php exec (needed for arc diff)<commit_after>{ stdenv, fetchgit, php, makeWrapper }:

let
  libphutil = fetchgit {
    url    = "git://github.com/facebook/libphutil.git";
    rev    = "c4cb6d99c4a5903079350f85fcc71895c0a0ea14";
    sha256 = "a7293aac4fdcfbaead09ee3e6ffb54c5d100b07905b4006194067411061ff994";
  };
  arcanist = fetchgit {
    url    = "git://github.com/facebook/arcanist.git";
    rev    = "50caec620a8ed45c54323cb71fee72fd0d935115";
    sha256 = "dd18ed22375ad1ba058703952be0d339d9c93704e9d75dd7e4e6625236dfe9b0";
  };
in
stdenv.mkDerivation rec {
  name    = "arcanist-${version}";
  version = "20140530";

  src = [ arcanist libphutil ];
  buildInputs = [ php makeWrapper ];

  unpackPhase = "true";
  buildPhase = "true";
  installPhase = ''
    mkdir -p $out/bin $out/libexec
    cp -R ${libphutil} $out/libexec/libphutil
    cp -R ${arcanist}  $out/libexec/arcanist

    ln -s $out/libexec/arcanist/bin/arc $out/bin

    wrapProgram $out/bin/arc \
      --prefix PATH : "${php}/bin"
  '';

  meta = {
    description = "Command line interface to Phabricator";
    homepage    = "http://phabricator.org";
    license     = stdenv.lib.licenses.asl20;
    platforms   = stdenv.lib.platforms.unix;
    maintainers = [ stdenv.lib.maintainers.thoughtpolice ];
  };
}

{'fraction_non_alphanumeric': 0.12920210611583638, 'fraction_numerical': 0.11826650465775618, 'mean_word_length': 3.363957597173145, 'pattern_counts': {'":': 0, '<': 3, '<?xml version=': 0, '>': 3, 'https://': 0, 'lorem ipsum': 0, 'www.': 0, 'xml': 0}, 'pii_count': 0, 'substrings_counts': 0, 'word_list_counts': {'cursed_substrings.json': 0, 'profanity_word_list.json': 0, 'sexual_word_list.json': 0, 'zh_pornsignals.json': 0}}
{'dir': 'git-commits-cleaned', 'id': '7582422', 'n_tokens_mistral': 1176, 'n_tokens_neox': 1004, 'n_words': 152}